
1. Présentation du projet et des objectifs de l’étude. 
1.1. CAST3M 

CAST3M est un logiciel de calcul aux éléments finis. Dans ce rapport, nous ne détaillerons pas les 

principes physiques de la méthode aux éléments finis. Vous pourrez trouver tous les éléments de 

réponse dans le cours Introduction à la méthode des éléments finis ou sur le site http://www-

cast3m.cea.fr/ . 

1.2. Situation du projet 

Le bâtiment étudié est situé au centre-ville de Bayonne est à une vocation de bureau. La structure est 

de type poteaux-poutres régulière. 

Notre objectif est de le transformer en complexe d’activités, ce qui implique de multiples 

changements, notamment au niveau de la structure.  

Veuillez trouver ci-après les principales transformations envisagées, représentées sur le logiciel Arche 

Ossature. 

  
Figure 1 Structure originale 

 
Figure 2 Projet à l'étude 

1.3. Objectifs de l’étude  

Notre objectif durant ce module d’initiation au calcul des éléments finis est d’analyser le 

comportement de la nouvelle structure sollicitée par : 

- Les charges de poids propre et d’exploitation, conformément pondérées avec la 

réglementation en vigueur (EuroCodes) : 1.35G + 1.5Q 

- Durant un séisme, dont nous détaillerons les caractéristiques dans ce rapport. 

Dans la suite nous détaillerons les étapes que nous avons entrepris pour résoudre ce problème.  

 

2. Détail du script et de la procédure de calcul 
2.1. Options générales du modèle 

dens1=1.; 
dens dens1; 
opti dime 3 elem cub8 TRAC OPEN EPSI LINEAIRE; 

DENS définit la densité générale du modèle (l’espacement des points). 

Il doit être adapté à chaque modèle. Ici une densite de 1m est 

cohérente. 

OPTI Définit les options générales de calcul. 

 

2.2. Détail de la géométrie  : les éléments du modèle 

2.2.1. Création de tous les éléments : utilisation des boucles 

Notre bâtiment est une répétition de portiques (poteaux / poutres), identiques. Pour optimiser le 

tracé, nous avons utilisé des boucles. Cette technique a aussi l’avantage de ne définir qu’une seule 

fois les éléments, et de s’assurer de l’unicité des points. 

Dans un premier temps nous avons donc créée pour 1 étage : 

- La dalle du sol 

- Les poutres de la dalle en question 

- Les poteaux de l’étage 

LLX=PROG 11 * 3.3; 
LLY=PROG 0.55 4.7 0.94 3.12 0.94 4.7 0.55; 
LLZ=PROG 7 * 3; 
P0=0. (-1 * (extr lly 1)) 0.; 
NX = DIME LLX; 
NY = DIME LLY; 
NZ = DIME LLZ; 
DDALL = P0; 
DPOUTR = P0; 
PI=P0; 
PTOT=PI; 
LZ=EXTR LLZ 1; 
VECZ=0. 0. LZ; 
LY= EXTr LLY 1; 
VECY= 0. LY 0.; 
PJ=PI PLUS VECZ; 
POTI=D PI PJ; 
POTOT=POTI; 
 
REPETER BOUX NX; 
      LX = EXTR LLX &BOUX; 
      VECX=LX 0. 0.; 

PROG permet de rentrer une liste de valeur 

 

DIME est un opérateur qui permet de trouver la dimension d’une liste 

 

EXTR permet de chercher une valeur précise au sein d’une liste 

 

D permet de tracer une droite entre deux points 

 

• Pour introduire une boucle : 

REPETER ‘Nom de la boucle’ ‘Nombre d’itérations’ ; 

• Pour terminer une boucle : 

FIN ‘Nom de la boucle’ 

      PI=PI PLUS VECX; 
       
      POTI=POTI PLUS VECX; 
      DDALL = DROITE DDALL PI; 
 
      PTOT=PTOT ET PI; 
      POTOT=POTOT ET POTI; 
      SDALL=DDALL; 
       
      PTOT2=PTOT; 
      POTOT2=POTOT; 
FIN BOUX;  
 
nbp=ptot2 nbno; 
tabpout=table; 
repeter boupout nbp; 
     tabpout.&boupout = ptot2 
point&boupout; 
fin boupout; 
 
REPETER BOUY NY;    
    LY= EXTr LLY &BOUY; 
    VECY= 0. LY 0.;        
    SDALL=SDALL TRANS VECY; 
    PTOT2=PTOT2 PLUS VECY;  
        
    repeter boupout nbp; 
         tabpout.&boupout = 
Dtabpout.&boupout (ptot2 point &boupout); 
fin boupout; 
    
    POTOT2=POTOT2 PLUS VECY; 
    POTOT=POTOT ET POTOT2;  
    PTOT=PTOT ET PTOT2; 
FIN BOUY; 
 
poutot=tabpout.1; 
repeter boupout (nbp - 1); 
     poutot=poutot et tabpout.(&boupout +1); 
fin boupout ;  

NBNO permet d’obtenir le nombre de point d’une géométrie 

 

TRANS crée une surface à partir d’une droite, en la translatant selon 

un vecteur. 

 

PLUS copie un élément à selon un vecteur donné 

Le résultat obtenu est le suivant : 

 

Les poteaux sont élevés, et l’ensemble est ensuite dupliqué sur les étages. 

spoutot=table; 
SDALLTOT=TABLE; 
I=1; 
spoutot.I=poutot; 
POUTRES=spoutot.1; 
SDALLTOT.I=SDALL; 
DALTOT=SDALLTOT.1; 

TABLE crée une table. Les éléments sont alors rangés successivement dans 

cette table. 

potot0=potot; 
REPETER BOUZ NZ;  
    POTOT0= POTOT0 plus  VECZ; 
    POTOT=POTOT ET POTOT0; 
    poutotj=spoutot.I plus vecz; 
    spoutot.(I+1)=poutotj; 
    SDALLJ=SDALLTOT.I plus VECZ; 
    SDALLTOT.(I+1)=SDALLJ; 
    I=I+1; 
    POUTRES=(POUTRES et poutotj) coul 
rouge; 
    DALTOT=DALTOT ET SDALLJ;   
FIN BOUZ;  

 

Nous obtenons ainsi : 

 

Il reste à créer la dalle du toit et les poutres qui lui sont associées. 

** DALLES 
ch1=manu chml sdalltot.2 scal 1.; 
x y z=coord ch1; 
sdall2=x elem INFE (5*3.3); 
 
sdalltot.2=sdall2; 
 
I=1; 
DALTOT=SDALLTOT.I; 
REPETER BOUZ (NZ);  
    I=I+1; 
    Si (I NEG 7); 
      DALTOT=DALTOT ET SDALLTOT.I;  
    FINSI;  
FIN BOUZ; 
DALTOIT=SDALLTOT.(NZ+1); 
 
 
** POUTRES 
ch1=manu chml spoutot.2 scal 1.; 
x y z=coord ch1; 
poutres2=x elem INFE (6*3.3); 

• Pour sélectionner des éléments d’une géométrie 

MANU ‘chml’ crée un champ manuel 

X Y Z sont les paramètres de ce champ 

Aller chercher les éléments voulus 

 

Dans la boucle BOUZ : 

NEG signifie « Non Egal » 

 

COUL attribue une couleur aux objets 

spoutot.2=poutres2; 
I=1; 
poutres=spoutot.I; 
REPETER BOUZ (NZ); 
   I=I+1; 
   Si (I NEG 7); 
     poutres=(poutres et spoutot.I) coul rouge; 
   Finsi; 
Fin BOUZ; 
Poutr8=spoutot.(NZ+1) coul rouge; 

 

2.2.2. Suppression des éléments inutiles 

Il faut maintenant supprimer les éléments qui ne sont pas présents dans le modèle final. 

La stratégie est la suivante :  

- Créer un champ qui sélectionne les éléments à supprimer. 

- Les soustraire au modèle initial. 

Pour le R-1 et le RDC : 

CH1=MANU CHML POTOT SCAL 1.; 
X Y Z=COORD CH1; 
POT1=Z ELEM COMPRIS (0.) (2*(LZ)) STRICTEMENT; 
 
CH1=MANU CHML POT1 SCAL 1.; 
X Y Z= COORD CH1; 
POT0=Y ELEM MINI; 
POTF=Y ELEM MAXI; 
POT3=X ELEM COMPRIS ((6*3.3)) ((10*3.3)); 
 
CH1=MANU CHML POT3 SCAL 1.; 
X Y Z= COORD CH1; 
POT4=Y ELEM COMPRIS (4.) (11.); 
 
CH1=MANU CHML POT1 SCAL 1.; 
X Y Z= COORD CH1; 
POT5=X ELEM COMPRIS (2.) (5*3.3+1); 
 
CH1=MANU CHML POT5 SCAL 1.; 
X Y Z= COORD CH1; 
POT6=Y ELEM COMPRIS (2.) (5.3); 
POT7=Y ELEM COMPRIS 9.24 12.; 
 
CH1=MANU CHML POT1 SCAL 1.; 
X Y Z= COORD CH1; 
POT8=X ELEM MINI; 
 
CH1=MANU CHML POT8 SCAL 1.; 
X Y Z= COORD CH1; 
POT10=Y ELEM COMPRIS (5.5) (9.30); 
 
CH1=MANU CHML POT1 SCAL 1.; 
X Y Z= COORD CH1; 
POT9=X ELEM MAXI; 
 
CH1=MANU CHML POT9 SCAL 1.; 
X Y Z= COORD CH1; 
POT11=Y ELEM COMPRIS (5.5) (9.30); 

pint1= pot11 et pot10 et pot0 et potf et pot4 et pot6 et pot7; 
*pint1 : poteau à éliminer au r-1 

 

Pour les autres étages, la procédure est exactement la même, veuillez trouver le détail en Annexe 4 : 

Script intégral. 

Une fois que les poteaux à éliminer ont été isolé, il faut les retirer du modèle. 

ppotot=chan potot poi1; 
pint=chan (pint1 et pint2 et pint3 et pint4) poi1; 
ppotf1=diff ppotot pint;  
potf1= potot elem appuye strictement ppotf1 coul vert;  

DIFF fait la différence d’objets entre deux géométrie 

 

Nous avons également modélisé la trémie d’escalier dans les étages courants. C’est un élément 

important car il apporte une fragilité locale qui peut fortement influencer le comportement global du 

bâtiment sous les effets d’un séisme. 

Nous avons d’ailleurs renforcé les bordures de la trémie pour améliorer ses performances. 

** Tremie escalier 
CH1=MANU CHML daltot SCAL 1.; 
X Y Z=COORD CH1; 
daltot=Z ELEM COMPRIS (-1.) (16) STRICTEMENT; 
 
ch1=manu chml daltot scal 1.; 
x y z=coord ch1; 
MASQ1=MASQ X COMPRIS (6.6) (9.9); 
MASQ2=MASQ Y COMPRIS (9.81) (14.68); 
masq3=masq1*masq2;  
tremis=masq3 elem infe 0.5; 
trou=masq3 elem supe 0.5; 
 
tremis = tremis et daltoit; 
 
** Renforcement  
trou=masq3 elem supe 0.5;  
poutrou=contour trou; 
poutres=poutres et poutrou; 
poutrTOT =poutres ; 
tout=potf1 et tremis et poutrTOT; 
 

MASQ agit comme un filtre 

 

Rq : La dalle du toit (daltoit) n’est pas concernée par 

cette trémie. 

 

Pour le renforcement on a cherché le contour, puis lui 

avons attribué les mêmes caractéristiques que nos 

poutres. 

 

 

2.3. Différenciation des éléments selon leur nature  : Charpente métallique Vs Béton 

armé 

Jusqu’à présent, nous avons supprimé des poteaux et des poutres qui participaient au maintien 

global de la structure. Nous devons donc renforcer les portiques modifiés. 

Nous avons choisi de réaliser des portiques en charpente métallique (poteaux renforcés + poutres 

moisées) à l’intérieur des portiques béton existant. 

Pour modéliser cette solution technique dans la géométrie CAST3M, nous avons recréée des poteaux 

à partir des poteaux déjà existants (cf. chapitre précédent). Ainsi, nous pourrons ensuite leur 

attribuer des caractéristiques différentes par la suite. 

Pour ce faire, nous avons suivi le protocole suivant : 

- Sélection des poteaux (ou poutres) sujets au renforcement 

- Attribution d’un nom et d’une couleur spécifique 

** CREATION DES POTEAUX RENFORCES 
CH1=MANU CHML POTf1 SCAL 1.; 
X Y Z=COORD CH1; 
POTr1=X ELEM COMPRIS (6*3.30-1) (11*3.30+1) STRICTEMENT; 
 
CH1=MANU CHML POTr1 SCAL 1.; 
X Y Z= COORD CH1; 
POTri=Y ELEM compris (-1) (1) coul bleu; 
POTrs=Y ELEM compris (14) (16) coul bleu; 
*trac (potri et potrs); 
*opti donn 5; 
 
CH1=MANU CHML POTf1 SCAL 1.; 
X Y Z=COORD CH1; 
POTr1=Z ELEM COMPRIS (5*3.) (21) STRICTEMENT; 
 
CH1=MANU CHML POTr1 SCAL 1.; 
X Y Z= COORD CH1; 
POTrx=x ELEM compris (4*3.30-1) (5*3.30+1); 
*trac potrx; 
 
CH1=MANU CHML POTrx SCAL 1.; 
X Y Z= COORD CH1; 
POTrxi=Y ELEM MINI coul bleu; 
POTrxs=Y ELEM MAXI coul bleu; 
 
potrenfo=potri et potrs et potrxi et potrxs; 
 
CH1=MANU CHML POTrenfo SCAL 1.; 
X Y Z= COORD CH1; 
POTrinf=Y ELEM compris -1 1 coul violet; 
POTrsup=Y ELEM compris 13 16 coul bleu; 

 

 

La même opération est nécessaire pour les poutres moisées. Le principe est identique et vous pouvez 

le consulter dans Annexe 4 : Script intégral entre les lignes 393 et 412. 

Nous obtenons ainsi les portiques renforcés suivant :  

Et la géométrie finale est la suivante : 

toutf= potf1 et tremis et poutres et poutr8 et poutif et pouti1d et potrsup et potrinf; 

 

 

 

2.4. Création des matrices de rigidité et de masse  

Afin de créer la matrice de rigidité il faut renseigner 2 champs importants :  

- Les hypothèses de calcul qui s’appliquent à l’élément sélectionné grâce à l’opérateur MODE 

- Les caractéristiques qui correspondent avec les hypothèses, grâce à l’opérateur MATE. 

Nous allons expliquer ici 2 exemples différents à savoir le traitement des dalles, et des poutres. Les 

matériaux utilisés dans notre modèle sont les suivants. 

*beton  
RHO_B=2400.; 
YG_B=30.E9; 
NU_B=0.2; 
EP=0.2; 
 
*acier 
RHO_A=7800.; 
YG_A=210.E9; 
NU_A=0.3; 
sig_a=235E6; 

RHO : masse volumique (kg/m3) 

YG : module d’Young (Pa) 

NU : Coefficient de Poisson 

EP : épaisseur de la dalle 

 

SIG : limite élastique (Pa) 

 

Les poteaux sont des éléments linéiques. 

** caracteristiques poteaux IPE bleu :potrsup 
modpot2=mode potrsup mecanique elastique plastique parfait poutre; 
 
*caract. poteaux CM 1* [152*152*23] 
sect4= 29.2*10e-4; 
tors4= 4.69*10e-8; 
inry4= 1250*10e-8; 
inrz4= 399.90*10e-8; 
matpot2=mate modpot2 SIGY sig_a YOUN yg_a NU nu_a RHO rho_a SECT sect4 
INRY inry4 INRZ inrz4 TORS tors4 DZ -0.2375; 
 
RIG5= RIGI modpot2 matpot2; 

MODE caractérise le modèle attribué à 

l’élément. 

Ici les poteaux renforcés sont en acier 

S235. Le modèle mécanique est donc 

mécanique élastique plastique parfait. 

Poutre renseigne sur le type d’élément. 

mas5= mass modpot2 matpot2; MATE définit les valeurs des champs 

induits dans MODE. 

 

Les dalles sont des éléments de surface. 

** caracteristiques dalles beton : tremis 
moddal= MODELE tremis MECANIQUE ELASTIQUE COQ4 COQ3; 
 
*caract; geometriques 
ep2=0.2; 
matdal= MATE moddal YOUN yg_b NU nu_b EPAI EP2 RHo rho_b; 
 
rig7= RIGI moddal matdal; 
mas7= mass moddal matdal; 

Dans ce cas le matériau est du béton et le type 

d’élément n’est plus une poutre mais une 

coque. 

 

La définition des autres rigidités et masses est disponible dans Annexe 4 : Script intégral. 

2.5. Définition des conditions aux limites  

Notre unique CL est le blocage de tous déplacements ou rotations de la dalle première dalle. En 

faisant cette hypothèse, nous négligeons les différences de comportement entre l’infrastructure (1 

niveau de sous-sol) et la superstructure.  

La stratégie est la suivante : 

- Isoler la dalle du R-1 

- Bloquer les rotations et déplacements 

** Condition aux limites 
 
CH1=MANU CHML tremis SCAL 1.; 
X Y Z= COORD CH1; 
base1=Z ELEM MINI; 
 
CL1= BLOQ depl ROTA (base1); 

BLOQ est l’opérateur qui permet de bloquer des déplacements 

et rotations. 

 

2.6. Calcul des modes propres de vibration du bâtiment  

Nous cherchons à déterminer la déformée « stable » du bâtiment lorsqu’il est soumis à une vibration 

continue, appelée fréquence propre. Ce paramètre dépend de la rigidité du bâtiment, et des masses 

propres. 

RIGtot1= RIGtot ET CL1; 
 
mastot= mas1 et mas2 et mas3 et mas4 et mas5 et mas6 et 
mas7; 
mastot= lump mastot; 
 
SOL1 = VIBRATION 'INTERVALLE' 0.1 20. BASSE 10  MAStot 
RIGtot1; 
 
N1=(DIME SOL1.MODES) - 2; 
REPETER BOU1 N1; 
     DEF1= DEFORMEE toutf SOL1.MODES. &BOU1 . 
DEFORMEE_MODALE; 
     trac DEF1 cach; 
FIN BOU1; 

SOL1 contient toutes les solutions pour lesquelles notre 

bâtiment rentre en vibration selon ses modes propres. 

 

VIBRATION est l’opérateur qui recherche ces solutions. 

L’option ‘Intervalle’ indique que CAST3M doit rechercher des 

solutions dans la plage de fréquence 0.1 à 20 Hz (car entre 

ces valeurs les séismes sont particulièrement préjudiciables 

pour les bâtiments), et qu’il doit renvoyer 10 résultats. 

 

Vous trouverez les différents modes propres du bâtiment dans le chapitre suivant, Analyse et 

interprétation des résultats. 

2.7.  Détail de la procédure Pas à Pas  

* Procedure      
DEBP EVDEP Point1*POINT TAB1*TABLE MOT1*MOT; 
      N1=DIME TAB1.TEMPS; 
      PROGT=PROG; 
      PROGD=PROG; 
      REPETER BOU1 N1; 
           IT=&BOU1 - 1; 
           TPS1=TAB1 . TEMPS . IT; 
           Droit1=EXTR TAB1 . DEPLACEMENTS . IT Point1 MOT1; 
           PROGT=INSE PROGT &BOU1 TPS1; 
           PROGD=INSE PROGD &BOU1 Droit1; 
      FIN BOU1; 
      cordo=chaine 'Déplacement ' MOT1 ' (m)'; 
      EVOL1=EVOL MANU 'Temps (S)' PROGT CORDO PROGD; 
FINP EVOL1; 

 

 

2.8.  Paramétrages des éléments externes à la structure  

2.8.1. Données du séisme 

Pour cette simulation nous avons utilisé l’accélérogramme d’un séisme réel, enregistré à Nice dont 

l’accélération maximale est de 2.5m/s2. 

CAST3M a recrée l’accélération à partir de deux fichiers que nous lui avons fournis : la valeur de 

l’accélération, au temps t.  

OPTI ACQU 'temps.list'; 
ACQU PROGT*LISTREEL 2048; 
OPTI ACQU 'accel.list'; 
ACQU PROGA*LISTREEL 2048; 
EVOLA=EVOL MANU PROGT PROGA; 

ACQU signifie acquérir un fichier externe 

 

EVOL reproduit l’évolution de l’accélération du séisme 

Le tracé de l’accélérogramme est disponible en Annexe 1 : Accélérogramme. 

 

2.8.2. Définition des champs de chargement 

• Le poids propre 

*Poids propre 
CH1=MANU 'CHPO' toutf UZ -9.81; 
PP=MAStot * CH1; 

Rq : Champ de pesanteur assimilé à tous les éléments du modèle. 

 

• Les charges d’exploitation 

FEXPL=-2500.; 
PRESS1=PRESSION 'COQU' moddal FEXPL 'NORM'; 

Champ de chargement appliqué à toutes les dalles. 

 

Les charges verticales totales sont pondérées à l’ELU. Nous pouvons créer le champ de chargement 

associé. De plus, nous avons calculé la déformée et le champ de contrainte développés uniquement 

par ce chargement. 

* Pondération des charges 1.35G + 1.5Q 
FTOT=(1.5 * PRESS1) + (1.35 * PP); 
 
* Définition du champ de chargement dû aux charges verticales 
prog1=prog 0. 2047.; 
prog2=prog 1. 1.; 

DEP1 : Déformée dû au chargement. 

 

SIG1 : Etat de contrainte dû au chargement. 

evol2=evol manu prog1 prog2; 
char2=chargement ftot evol2 'MECA'; 
 
* Déplacements 
DEP1=RESOU RIGTOT1 FTOT; 
 
* Contraintes  
SIG1=SIGMA DEP1 MODTOT MATTOT; 

 

• Les charges sismiques 

* Création du champ de chargement dû au séisme 
CHX=MANU CHPO toutf 'UY' 1.; 
FX=-1.*MAStot*CHX; 
char1=chargement FX EVOLA 'MECA'; 

 

 

2.9. Paramétrage de la procédure Pas à Pas  

Les procédures Pas à Pas servent à résoudre des problèmes transitoires. Dans notre cas, un séisme, 

le pas est temporel. CAST3M va donc calculer les champs de déplacements et de contraintes à 

chaque instant séparés d’un intervalle, nommé PAS1. Dans cette simulation, il y a en tout 2048 pas 

de 7.8125E10-3 secondes chacun, soit un séisme de 16 secondes. 

Un autre point important est de préciser qu’au temps t=0, le bâtiment est déjà chargé selon la 

combinaison 1.35G+1.5Q. Ainsi, lorsque la procédure commence le bâtiment est déjà déformé et 

soumis à un état de contrainte. Si on ne le fait pas, le chargement brusque couplé au séisme 

provoquerait des vibrations immédiates, non représentatives de la réalité. 

* Paramètres procédure Pas à Pas 
PAS1=7.8125E-03; 
TAB1=TABLE; 
TAB1.DEPLACEMENTS=TABLE; 
TAB1.DEPLACEMENTS . 0=DEP1; 
TAB1.CONTRAINTES=TABLE; 
TAB1.CONTRAINTES . 0 = SIG1; 
TAB1.MODELE=MODtot; 
TAB1.CARACTERISTIQUES=MAtTOT; 
TAB1.DYNAMIQUE=VRAI; 
TAB1.CHARGEMENT=CHAR1; 
TAB1.TEMPS_CALCULES=PROG 0. PAS PAS1 NPAS 2047; 
TAB1.BLOCAGES_MECANIQUES=CL1; 
PASAPAS TAB1; 

- Pour imposer la déformée au temps t=0 : 

TAB1.DEPLACEMENTS . 0=DEP1; 

 

- Pour imposer les contraintes au temps t=0 : 

TAB1.CONTRAINTES . 0 = SIG1; 

 

Cette étape est très lourde en puissance de calcul, c’est pourquoi à l’issue il est préférable de réaliser 

une sauvegarde. 

2.10. Sauvegarde du calcul  

A titre d’exemple, la sauvegarde totale de notre calcul occupait un fichier de 3.3 Go, d’où 

l’importance de bien sauvegarder (sur un disque dur de préférence, pour ne pas perdre de temps en 

vitesse de transfert). 

*sauvegarde 
OPTI SAUV 'IBAI_ARTEAN_vf.sauv'; 
sauv; 

Le fichier ‘IBAI_ARTEAN_vf.sauv’ s’enregistre automatiquement dans le 

même répertoire que le script CAST3M. 

Attention, il vaut mieux le mettre sur le bureau au préalable. (Problème 

d’enregistrement si directement sur le DD). 

2.11. Extraction des résultats  

2.11.1. Modélisation dynamique du séisme 

opti trac 'X'; 
@cartoon tab1 toutf (0. 1.e3 0.); 
@excel1 (EV0y) 'EV0y'; 

Rq : L’affichage est long à charger. 

Pour avoir un rendu dynamique de notre bâtiment soumis au séisme 

 

2.11.2. Recherche du pas pour lequel l’amplitude de la déformée est maximale 

La simulation complète dure 16 secondes. Durant ce laps de temps, le bâtiment bouge énormément 

et passe par de nombreux états de contraintes et de déformations différents. 

Nous avons fixé comme référence un point situé à l’angle de la plus haute dalle. 

Nous pouvons tracer l’évolution de ce point dans l’espace selon x,y et z. 

* Choix du point d’étude 
vecz2= (3.*6); 
point0=toutf point proch (p0 plus (0. 0. vecz2)); 
 
* Evolution spaciale du point selon X et Y 
EV0x=EVDEP Point0 TAB1 'UX'; 
ev0y=EVDEP Point0 TAB1 'UY'; 
 
* Tracé de l’amplitude du déplacement 
DESS (EV0x et ev0y); 

Rq : L’affichage est long à charger. 

 

Le graphique suivant nous permet d’obtenir le pas pour lequel le déplacement est maximum. Il s’agit 

du pas n° 1157, à t = 8,6 s. 

 

 

 

On peut également voir sur ce graphe que le bâtiment passe par 3 phases de vibration. 

- La première quand le bâtiment commence à vibrer. Entre t = 0s à t = 4s. 

- La deuxième quand le bâtiment atteint son pic d’excitation. Entre t = 4s à t = 8s. 

- La troisième quand les oscillations du bâtiment se stabilisent. Entre t = 8s jusqu’à la fin. 

En Annexe 2 et 3 vous trouverez les tracer indépendants des déplacements selon l’axe x et selon l’axe 

y respectivement.  

 

2.11.3. Extraction des contraintes  

Dans le cas des éléments linéiques (poutres, poteaux) : 

trac tab1.contraintes.1157 modpou matpou;  

 

Cette commande permet d’obtenir les efforts normaux, efforts tranchants et moments fléchissants 

dans la poutre. (Cf Analyse et interprétation des résultats) 

Dans le cas des éléments surfaciques, on se réfère aux contraintes de VonMisess. Or il est impossible 

de savoir si nos dalles sont en traction ou en compression en utilisant ce critère. 

Nous allons donc isoler la surface supérieure ou inférieure de la dalle. 

CHAM2 = CALP tab1.contraintes.1157  moddal matdal infe ;  
 trac cham2 moddal; 

 

 

Vous trouverez les résultats dans le chapitre suivant, Analyse et interprétation des résultats. 

 

2.11.4. Extraction des déformations 

trac tab1.deplacements.1157 poutres;  

 

 

 

 

 

 

 

 

 

 

3. Analyse et interprétation des résultats 
3.1. Analyse des différents modes propres. 

3.1.1. Premier mode propre : 1.065 Hz 

Les premiers modes nous révèlent les faiblesses du bâtiment. 

Dans ce cas précis, il manque de contreventement dans le sens de la longueur (axe X). 

Nous proposons de contreventer les portiques à l’aide de croix de saint-andré, par exemple 

 

3.1.2. Deuxième mode propre : 1.239 Hz 

 

Au deuxième mode propre il apparait de la torsion.  

Les poteaux du RDC sont principalement touchés à cause de leur élancement (plus grand que ceux 

des étages) et du poids qu’ils reprennent. 

Il faudra surveiller leur comportement en contraintes et déformations. 

 

3.1.3.  Troisième mode propre : 2.204 Hz 

Mode propre similaire au précédent. 

Mêmes préconisations. 

 

3.1.4.  Quatrième mode propre : 2.581 Hz 

 

La rigidité et l’élancement des files de poteaux sont défaillants dans ce cas de figure. 

Nous préconisons un contreventement vertical pour rigidifier les portiques entre eux. 

 

3.1.5.  Cinquième mode propre : 3.141 Hz 

La rigidité des portiques est mise à mal par ce mode propre. 

Nous préconisons de les contreventer en façade. 

 

3.1.6.  Sixième mode propre : 4.239 Hz 

 

Apparition d’une défaillance locale au niveau de la dalle du toit. 

Cela est dû à la suppression des poteaux en travée au R+4. 

Le comportement de la dalle sera approfondi dans l’étude des contraintes et des déformations. 

 

3.1.7.  Septième mode propre : 4.851 Hz 

Idem que sixième mode propre. 

 

3.1.8.  Huitième mode propre : 4.874 Hz 

 

Combinaison des modes propres précédents. 

 

 

 

 

3.1.9.  Neuvième mode propre : 5.008 Hz 

Combinaison des modes propres précédents. 

 

3.1.10. Dixième mode propre : 5.304 Hz 

 

Combinaison des modes propres précédents. 

 

 

 

 

 

3.2. Analyse des contraintes et des déplacements éléments par éléments  

3.2.1.  Poteaux existants et conservés, en BA  

• Contraintes  

 

Figure 3 Représentation des contraintes dans les poteaux BA. 

Pour les contraintes des poteaux nous nous concentrons seulement sur l’effort normal. 

 σadm,beton σmax 

Compression = - 16.67 MPa = - 1.84 MPa 

Traction = + 3.33 MPa = + 1.95 MPa 

  ➔ Dimensionnement OK. 

 

 

• Déplacements  

L’axe x est orienté dans la longueur du bâtiment. 

L’axe y est orienté dans la largeur du bâtiment. 

Nous rappelons que les déplacements verticaux (axe z) ne sont pas à prendre en compte en France 

métropolitaine. 

Figure 4 Représentation des déplacements des poteaux BA suivant l'axe X. 

 

Figure 5 Représentation des déplacements des poteaux BA suivant l'axe Y. 

 

δadm δmax,x δmax,y 

= 37.5 mm = 65.4 mm = 128 mm 

  ➔ Dimensionnement A REVOIR. 

3.2.2.  Poutres existantes et renforcées, en BA  

• Contraintes  

 

Figure 6 Représentation des contraintes dans les poutres BA. 

Pour les contraintes des poutres nous nous concentrons seulement sur l’effort normal. 

 σadm,beton σmax 

Compression = - 16.67 MPa = - 0.198 MPa 

Traction = + 3.33 MPa = + 0.204 MPa 

  ➔ Dimensionnement OK. 

• Déplacements  

 

Figure 7 : Représentation des déplacements des poutres BA suivant l'axe X. 

 

Figure 8 : Représentation des déplacements des poutres BA suivant l'axe Y. 

δadm δmax,x δmax,y 

= 37.5 mm = 65.4 mm = 128 mm 

  ➔ Dimensionnement A REVOIR. 

3.2.3.  Dalles, en BA 

• Contraintes 

 

Figure 9 : Représentation des contraintes dans les dalles BA. 

 σadm,beton σmax 

Compression = - 16.67 MPa = - 16.6 MPa 

Traction = + 3.33 MPa = + 20 MPa 

  ➔ Dimensionnement A REVOIR. 

• Déplacements  

 

Figure 10 : Représentation des déplacements des dalle BA suivant l'axe X. 

 

Figure 11 : Représentation des déplacements des dalles BA suivant l'axe Y. 

δadm δmax,x δmax,y 

= 37.5 mm = 65.5 mm = 128 mm 

  ➔ Dimensionnement A REVOIR. 

3.2.4.  Poteaux renforcés, en acier S235 

• Contraintes  

 

Figure 12 : Représentation des contraintes dans les poteaux de renforcement en acier S235. 

 σadm,acier σmax 

Compression = - 235 MPa = - 0.585 MPa 

Traction = + 235 MPa = + 0.847 MPa 

  ➔ Dimensionnement OK. 

 

• Déplacements 

 



 
 

28 
 

 

Figure 13 :  Représentation des déplacements des poteaux de renforcement en acier S235 suivant l'axe X. 

 

 

Figure 14 : Représentation des déplacements des poteaux de renforcement en acier S235 sur l'axe Y. 

 

δadm δmax,x δmax,y 

= 37.5 mm = 65.4 mm = 128 mm 

  ➔ Dimensionnement A REVOIR. 

 

3.2.5.  Poutres renforcements, en acier S235 

• Contraintes  

 

Figure 15 : Représentation des contraintes dans les poutres de renforcement en acier S235. 

 σadm,acier σmax 

Traction = + 235 MPa = + 0.316 MPa 

Compression = - 235 MPa = - 0.326  MPa 

  ➔ Dimensionnement OK. 

• Déplacements  

 

Figure 16 : Représentation des déplacements des poutres de renforcement en acier S235, selon l'axe X. 

 

Figure 17 : Représentation des déplacements des poutres de renforcement en acier S235, selon l'axe Y. 

δadm  δmax,x δmax,y 

= 37.5 mm = 65.4 mm = 128 mm 

  ➔ Dimensionnement A REVOIR. 

4. Conclusion 
 

L’analyse aux éléments finis de la structure décrite conformément aux documents de référence a 

montré les résultats suivants : 

 Contraintes : σadm > σmax Déplacements : δadm > δmax 

Dalle BA A REVOIR A REVOIR 

Poutres BA OK A REVOIR 

Poteaux BA OK A REVOIR 

Poutres acier S235 OK A REVOIR 

Poteaux acier S235 OK A REVOIR 

 

Nous sommes donc en mesure de dire que cette structure n’est pas correctement dimensionnée vis-

à-vis des sollicitations sismiques + (poids propre et charge d’exploitation).  

5. Annexes  

5.1. Annexe 1 : Accélérogramme du séisme simulé  

5.2. Annexe 2 : Evolution du déplacement du point étudié durant le séisme, selon 

l’axe x.  

 

5.3. Annexe 3 : Evolution du déplacement du point étudié durant le séisme, selon 

l’axe y.  

 

5.4. Annexe 4 : Script intégral  

Cf fichier joint IBAI_ARTEAN_vf.dgibi  

 




